```
>
```

Exercice 4 Condition de diagonalisation

```
> restart: with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

```
> A:=matrix(5,5,[a,0,0,0,b,0,a,1,b,0,0,0,c,0,0,0,b,1,a,0,b,0,0,0,a]);
```

$$A := \begin{bmatrix} a & 0 & 0 & 0 & b \\ 0 & a & 1 & b & 0 \\ 0 & 0 & c & 0 & 0 \\ 0 & b & 1 & a & 0 \\ b & 0 & 0 & 0 & a \end{bmatrix}$$

```
> r:=eigenvects(A);
```

$$r := [b+a, 2, \{[1, 0, 0, 0, 1], [0, 1, 0, 1, 0]\}], [c, 1, \{[0, 1, c-a-b, 1, 0]\}], [-b+a, 2, \{[0, -1, 0, 1, 0], [-1, 0, 0, 0, 1]\}]$$

```
> P:=transpose(matrix(5,5,[op(r[1][3]),op(r[2][3]),op(r[3][3])]));
```

$$P := \begin{bmatrix} 1 & 0 & 0 & 0 & -1 \\ 0 & 1 & 1 & -1 & 0 \\ 0 & 0 & c-a-b & 0 & 0 \\ 0 & 1 & 1 & 1 & 0 \\ 1 & 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> B:=simplify(evalm(inverse(P)&*A&*P));
```

$$B := \begin{bmatrix} b+a & 0 & 0 & 0 & 0 \\ 0 & b+a & 0 & 0 & 0 \\ 0 & 0 & c & 0 & 0 \\ 0 & 0 & 0 & -b+a & 0 \\ 0 & 0 & 0 & 0 & -b+a \end{bmatrix}$$

```
>
```

Exercice 5 Valeur approchée d'une somme

```
> restart: s[2]:=evalf(ln(2)/4):
> for k from 2 to 100 while evalf(ln(k)/(k^2))>0.001 do s[k+1]:=s[k]+(-1)^(k+1)*ln(k+1)/((k+1)^2) od:
> (evalf(s[k-1],5), evalf(s[k],5), k);
```

$$0.10185, 0.10086, 65$$

```
>
```

Exercice 6 Détermination d'un commutant

```
> restart: with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

```
> A:=matrix(3,3,[a,b,c,b,a,b,c,c,a]);
```

$$A := \begin{bmatrix} a & b & c \\ b & a & b \\ c & c & a \end{bmatrix}$$

```
> M:=matrix(3,3,[seq(x[k],k=1..9)]);
```

$$M := \begin{bmatrix} x_1 & x_2 & x_3 \\ x_4 & x_5 & x_6 \\ x_7 & x_8 & x_9 \end{bmatrix}$$

```
> P:=evalm(A&*M-M&*A):
> eq:=(i,j)->P[i,j]=0;
```

$$eq := (i, j) \to P_{i,j} = 0$$

```
> r:=solve({seq(seq(eq(i,j),j=1..3),i=1..3)},{seq(x[k],k=1..9)});
```

$$r := \left\{ x_4 = \frac{-c\,x_1 + b\,x_8 + c\,x_9}{c}, x_1 = x_1, x_8 = x_8, x_7 = x_8, x_9 = x_9, x_6 = \frac{-c\,x_1 + b\,x_8 + c\,x_9}{c}, x_5 = \frac{b\,x_1 - c\,x_9 + c\,x_1}{b}, x_3 = -x_9 + x_1 + x_8, x_2 = \frac{b^2\,x_8 + b\,c\,x_9 - b\,c\,x_1 + c^2\,x_9 - c^2\,x_1}{c\,b} \right\}$$

```
> assign(r);
> x[1]:=x: x[8]:=y: x[9]:=z:
> S:=matrix(3,3,[seq(x[k],k=1..9)]);
```

$$S := \begin{bmatrix} x & \dfrac{b^2\,y + b\,c\,z - b\,c\,x + c^2\,z - c^2\,x}{c\,b} & -z + x + y \\ \dfrac{-c\,x + b\,y + c\,z}{c} & \dfrac{b\,x - c\,z + c\,x}{b} & \dfrac{-c\,x + b\,y + c\,z}{c} \\ y & y & z \end{bmatrix}$$

```
>
```

On conclut que si bc est non nul la dimension est 3

Exercice 7 Vecteurs propres fixés

```
> restart: with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

```
> P:=matrix(3,3,[1,1,1,1,2,1,1,1,-1]);
```

$$P := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 1 \\ 1 & 1 & -1 \end{bmatrix}$$

```
> det(P);
```

$$-2$$

```
> A:=matrix(3,3,[a,1,b,1,c,d,e,f,1]);
```

$$A := \begin{bmatrix} a & 1 & b \\ 1 & c & d \\ e & f & 1 \end{bmatrix}$$

```
> B:=evalm(inverse(P)&*A&*P):
```

```
> r:=solve({B[1,2]=0, B[1,3]=0, B[2,1]=0, B[2,3]=0, B[3,1]=0, B[3,2]=0},{a,b,c,d,e,f});
```

$$r := \{f = 1, b = -3, a = 0, e = -4, c = 0, d = -3\}$$

```
> assign(r):
```

```
> RA:=evalm(matrix(3,3,[a,1,b,1,c,d,e,f,1]));
```

$$RA := \begin{bmatrix} 0 & 1 & -3 \\ 1 & 0 & -3 \\ -4 & 1 & 1 \end{bmatrix}$$

```
> evalm(inverse(P)&*RA&*P);
```

$$\begin{bmatrix} -2 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 4 \end{bmatrix}$$

```
>
```

Exercice 8 condition de diagonalisation

```
> restart: with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

```
> A:=matrix(4,4,[1,0,0,0,a,1,0,0,b,c,2,0,d,e,f,2]);
```

$$A := \begin{bmatrix} 1 & 0 & 0 & 0 \\ a & 1 & 0 & 0 \\ b & c & 2 & 0 \\ d & e & f & 2 \end{bmatrix}$$

```
> eigenvects(A);
```

$$[2, 2, \{[0, 0, 0, 1]\}], [1, 2, \{[0, 1, -c, -e + f c]\}]$$

```
> Id:=diag(1,1,1,1):
```

```
> B:=evalm((A-Id)&*(A-2*Id));
```

$$B := \begin{bmatrix} 0 & 0 & 0 & 0 \\ -a & 0 & 0 & 0 \\ c\,a & 0 & 0 & 0 \\ e\,a + f\,b & f\,c & f & 0 \end{bmatrix}$$

>

A est diagonalisable si et seulement si B=0 d'où la condition nécessaire et suffisante a=f=0

Exercice 9

```
> restart: ps:=(P,Q)->sum((subs(x=k,P))*(subs(x=k,Q)), k=0..4);
```

$$ps := (P, Q) \to \sum_{k=0}^{4} subs(x = k, P)\, subs(x = k, Q)$$

```
> P0:=interp([0,1,2,3,4],[1,2,1,2,4],x);
```

$$P0 := -\frac{5}{24}x^4 + \frac{23}{12}x^3 - \frac{127}{24}x^2 + \frac{55}{12}x + 1$$

```
> P1:=sort(P0-a-b*x-c*x^2);
```

$$P1 := -\frac{5}{24}x^4 + \frac{23}{12}x^3 - x^2 c - b\,x - \frac{127}{24}x^2 - a + \frac{55}{12}x + 1$$

```
> eq:=i-> ps(P1,x^i)=0;
```

$$eq := i \to ps(P1, x^i) = 0$$

```
> r:=solve({eq(0),eq(1),eq(2)},{a,b,c});
```

$$r := \left\{ a = \frac{48}{35}, c = \frac{2}{7}, b = \frac{-19}{35} \right\}$$

```
> assign(r); resultat:= a+b*x+c*x^2;
```

$$resultat := \frac{48}{35} - \frac{19}{35}x + \frac{2}{7}x^2$$

>

Exercice 10 Condition pour qu'un polynôme ait une racine de module 1

```
> restart: P:=x^4+x^3+a*x^2+x+2;
```

$$P := x^4 + x^3 + a\,x^2 + x + 2$$

```
> Q:=sort(expand((x^2+b*x+c)*(x^2+d*x+1))):
> eq:=k->coeff(P,x,k)=coeff(Q,x,k):
> r:=solve({eq(0),eq(1),eq(3)},{b,c,d});
```

$$r := \{d = 0, b = 1, c = 2\}$$

```
> assign(r); solve(eq(2),a);
```

$$3$$

```
> solve(x^2+b*x+c,x);
```

$$-\frac{1}{2} + \frac{1}{2} I\sqrt{7}, -\frac{1}{2} - \frac{1}{2} I\sqrt{7}$$

```
> solve(x^2+d*x+1,x);
```

$I, -I$

```
>
```